Hierarchical Bayesian Modeling of the English Premier League

Milad Kharratzadeh

14 January, 2017

Contents

<table>
<thead>
<tr>
<th>Section</th>
<th>Page</th>
</tr>
</thead>
<tbody>
<tr>
<td>Introduction</td>
<td>2</td>
</tr>
<tr>
<td>Model</td>
<td>2</td>
</tr>
<tr>
<td>Reading and Munging the Data</td>
<td>3</td>
</tr>
<tr>
<td>Stan Code</td>
<td>4</td>
</tr>
<tr>
<td>Fitting the Model</td>
<td>5</td>
</tr>
<tr>
<td>Evolution of Team Abilities</td>
<td>7</td>
</tr>
<tr>
<td>Parameter Estimates</td>
<td>10</td>
</tr>
<tr>
<td>Model Checking</td>
<td>10</td>
</tr>
<tr>
<td>Making Probabilistic Predictions with The Model</td>
<td>11</td>
</tr>
</tbody>
</table>
Introduction

In this case study, we provide a hierarchical Bayesian model for the English Premier League in the season of 2015/2016. The league consists of 20 teams and each two teams play two games with each other (home and away games). So, in total, there are 38 weeks, and 380 games. We model the score difference (home team goals − away team goals) in each match. The main parameters of the model are the teams’ abilities which is assumed to vary over the course of the 38 weeks. The initial abilities are determined by performance in the previous season plus some variation. Please see the next section for more details.

We implement and fit our model in Stan and prepare the data and analyze the results in R.

Model

The score difference in game i, denoted as $y_i$, is modeled as a $t$ distribution:

$$y_i \sim t_\nu(a_{\text{home}_i} - a_{\text{away}_i} + b_{\text{home}}, \sigma_y),$$

where $a_{w,j}$ is the ability of team $j$ in week $w$. Because of the irregularities in the schedule of the games, the ‘week’ for the home and away teams may not be the same; so, $\text{home}_i$ and $\text{away}_i$ denote the week for the home and away team in game $i$ respectively. The possible advantage (or disadvantage) for the home team is modeled by the variable $b_{\text{home}}$; we do not expect this effect to be large, and therefore, assign it a $N(0,1)$ weak prior. The variation in the score difference is modeled by $\sigma_y$ and we give it a weak prior $N(0,5)$. The degrees of freedom, $\nu$, has a prior of $\text{Gamma}(2,0.1)$.

We assume that the abilities of the teams can change during the season (due to injuries, player forms, etc.). We assume the following random walk model:

$$a_{w,j} \sim N(a_{w-1,j}, \sigma_{aj}), \quad w = 2, \ldots, 38$$

where $\sigma_{aj}$ models the game-to-game variability for team $j$. We assume a hierarchical structure where the variations in team abilities are sampled from $N(0,\tau_a)$, and $\tau_a \sim \text{Cauchy}(0,1)$.

The ability for the first week is derived from the previous season performance with some variability:

$$a_{1,j} \sim N(b_{\text{prev},a_0,j}, \sigma_{a0}),$$

where $b_{\text{prev}}$ is the regression coefficient and $a_{0,j}$ is a score between $-1$ and $1$ achieved by a linear transformation of the total points achieved from last season. We expect some team-level variation in the initial performance; this is modeled by $\sigma_{a0}$. Both $b_{\text{prev}}$ and $\sigma_{a0}$ have weakly informative priors, $N(0,5)$.

We fit our model every week using all the matches up until that point. (Therefore, we fit our model 38 times.)

---

1 As suggested by: Juarez and Steel, “Model-based clustering of non-Gaussian panel data based on skew-t distributions”, Journal of Business & Economic Statistics 28 (2010), 52-66
Reading and Munging the Data

We first read the data from the website football-data.co.uk and save it to a list called epl. The main components of this list are home_team, away_team, and score_diff which have 380 elements each. Teams have fixed IDs which are integers from 1 to 20 assigned to teams sorted alphabetically. The previous performance (points in previous season) is stored in a separate CSV file; this data is read and mapped to a score between $-1$ and $+1$ using the user-defined function map_to_score. The variable home_week, also of length 380, identifies the ‘week’ for the home team (i.e., how many matches the home team has played so far, including the current match).

```r
library(plyr)
# Linear map of points to a score between -1 and 1
map_to_score <- function(x) {
  x_max <- max(x); x_min <- min(x);
  return(2*x/(x_max-x_min) - (x_max+x_min)/(x_max-x_min))
}

url_csv <- "http://www.football-data.co.uk/mmz4281/1516/E0.csv";
# Data downloaded from football-data.co.uk
mydat <- read.csv(url(url_csv)); epl <- c();
# teams are assigned IDs 1, 2, ...

epl$home_team <- as.numeric(mydat$HomeTeam)
epl$away_team <- as.numeric(mydat$AwayTeam)
epl$team_names <- levels(mydat$HomeTeam)
epl$home_goals <- mydat$FTHG  # FTHG: full time home goals
epl$away_goals <- mydat$FTAG  # FTHG: full time away goals
epl$score_diff <- epl$home_goals - epl$away_goals
# Points from last season are read and mapped to a score

epl$prev_perf <- read.csv('DATA/prev_perf.csv', header = FALSE)
epl$prev_perf <- map_to_score(epl$prev_perf[,2])
epl$nteams <- length(unique(epl$home_team))
epl$ngames <- length(epl$score_diff)
epl$nweeks <- floor(2*epl$ngames/epl$nteams)
# The following code computes the week for each team in their games:

epl$home_week <- c(); epl$away_week <- c();
for (g in 1:epl$ngames) {
  epl$home_week[g] <- sum(epl$home_team[1:g] == epl$home_team[g]) + 
  sum(epl$away_team[1:g] == epl$home_team[g])
epl$away_week[g] <- sum(epl$away_team[1:g] == epl$away_team[g]) + 
  sum(epl$home_team[1:g] == epl$away_team[g])
}
epl$bet_home <- mydat$B365H;  # Betting odds for home team win
epl$bet_draw <- mydat$B365D;  # Betting odds for draw
epl$bet_away <- mydat$B365A;  # Betting odds for away team win
saveRDS(epl,'epl_data.rds')
```


## Stan Code

The Stan code for the model is shown below. The code is commented and self-explanatory.

In the generated quantities block, we sample replications data for the `score_diff`; we will use these later for posterior predictive checks.

```stan
data {
  int<lower=1> nteams; // number of teams (20)
  int<lower=1> ngames; // number of games
  int<lower=1> nweeks; // number of weeks
  int<lower=1> home_week[ngames]; // week number for the home team
  int<lower=1> away_week[ngames]; // week number for the away team
  int<lower=1, upper=nteams> home_team[ngames]; // home team ID (1, ..., 20)
  int<lower=1, upper=nteams> away_team[ngames]; // away team ID (1, ..., 20)
  vector[ngames] score_diff; // home_goals - away_goals
  row_vector[nteams] prev_perf; // a score between -1 and +1
}

parameters {
  real b_home; // the effect of hosting the game in mean of score_diff dist.
  real b_prev; // regression coefficient of prev_perf
  real<lower=0> sigma_a0; // teams ability variation
  real<lower=0> tau_a; // hyper-param for game-to-game variation
  real<lower=1> nu; // t-dist degree of freedom
  real<lower=0> sigma_y; // score_diff variation
  row_vector<lower=0>[nteams] sigma_a_raw; // game-to-game variation
  matrix[nweeks, nteams] eta_a; // random component
}

transformed parameters {
  matrix[nweeks, nteams] a; // team abilities
  row_vector<lower=0>[nteams] sigma_a; // game-to-game variation
  a[1] = b_prev * prev_perf + sigma_a0 * eta_a[1]; // initial abilities (at week 1)
  sigma_a = tau_a * sigma_a_raw;
  for (w in 2:nweeks) {
    a[w] = a[w-1] + sigma_a .* eta_a[w]; // evolution of abilities
  }
}

model {
  vector[ngames] a_diff;
  // Priors
  nu ~ gamma(2,0.1);
  b_prev ~ normal(0,1);
  sigma_a0 ~ normal(0,1);
}
```
sigma_y ~ normal(0,5);
b_home ~ normal(0,1);
sigma_a_raw ~ normal(0,1);
tau_a ~ cauchy(0,1);
to_vector(eta_a) ~ normal(0,1);

// Likelihood
for (g in 1:ngames) {
    a_diff[g] = a[home_week[g],home_team[g]] - a[away_week[g],away_team[g]];
}
score_diff ~ student_t(nu, a_diff + b_home, sigma_y);

} generated quantities {
    vector[ngames] score_diff_rep;
    for (g in 1:ngames)
        score_diff_rep[g] = student_t_rng(nu, a[home_week[g],home_team[g]] -
        a[away_week[g],away_team[g]]+b_home, sigma_y);
}

Fitting the Model

As mentioned earlier, we fit the model multiple times, after every 10 games. In the code below, epl_w contains all the data for matches in the first \( w \times 10 \) matches. We fit the model with 4 chains of length 750 (with the first half for warmup), for a total of 1500 samples (after warmup).

```r
library("rstan")
epl <- readRDS("epl_data.rds")
rstan_options(auto_write = TRUE)
options(mc.cores = parallel::detectCores())
sm <- stan_model("epl_model.stan")
nsamples <- 1500
a_sims <- array(NA, c(nsamples, epl$nweeks, epl$nteams))
for (w in 1:38) {
    epl_w <- epl
    idx <- c(1:(w*10))
epl_w$home_team <- epl$home_team[idx]epl_w$away_team <- epl$away_team[idx]epl_w$home_goals <- epl$home_goals[idx]epl_w$away_goals <- epl$away_goals[idx]epl_w$score_diff <- epl$score_diff[idx]epl_w$home_week <- epl$home_week[idx]
```
epl_w$away_week  <- epl$away_week[idx]
epl_w$ngames    <- w*10
epl_w$nweeks     <- max(c(epl_w$home_week, epl_w$away_week))
fit <- sampling(sm, chains = 4, iter = (nsamples/2), data = epl_w)
saveRDS(fit, paste("FITS/fit_", w, ".rds", sep=""))
sims <- extract(fit)
for (g in ((w-1)*10 + 1):(w*10)) {
  a_sims[,epl$home_week[g],epl$home_team[g]] <-
    sims$a[,epl$home_week[g],epl$home_team[g]]
  a_sims[,epl$away_week[g],epl$away_team[g]] <-
    sims$a[,epl$away_week[g],epl$away_team[g]]
}
saveRDS(a_sims,"FITS/a_sims.rds")
Evolution of Team Abilities

Remember that we re-fit the model after each week. In the first table below, we show the estimated team abilities (+/- 1 s.e.) after week 1. The teams in the table are sorted according to their performance in the previous season. We observe that after only one match for each team, the estimated abilities are somewhat similar to previous performance with perturbations due to the results in the first week. We also observe that the uncertainty intervals are quite wide; this also makes sense, because we have only one observation per team.

<table>
<thead>
<tr>
<th>Team abilities after week 1 (estimate +/- 1 s.e.)</th>
<th>Teams are sorted according to previous performance</th>
</tr>
</thead>
<tbody>
<tr>
<td>Chelsea</td>
<td></td>
</tr>
<tr>
<td>Man City</td>
<td></td>
</tr>
<tr>
<td>Arsenal</td>
<td></td>
</tr>
<tr>
<td>Man United</td>
<td></td>
</tr>
<tr>
<td>Tottenham</td>
<td></td>
</tr>
<tr>
<td>Liverpool</td>
<td></td>
</tr>
<tr>
<td>Southampton</td>
<td></td>
</tr>
<tr>
<td>Swansea</td>
<td></td>
</tr>
<tr>
<td>Stoke</td>
<td></td>
</tr>
<tr>
<td>Crystal Palace</td>
<td></td>
</tr>
<tr>
<td>West Ham</td>
<td></td>
</tr>
<tr>
<td>Everton</td>
<td></td>
</tr>
<tr>
<td>West Brom</td>
<td></td>
</tr>
<tr>
<td>Leicester</td>
<td></td>
</tr>
<tr>
<td>Newcastle</td>
<td></td>
</tr>
<tr>
<td>Sunderland</td>
<td></td>
</tr>
<tr>
<td>Aston Villa</td>
<td></td>
</tr>
<tr>
<td>Watford</td>
<td></td>
</tr>
<tr>
<td>Norwich</td>
<td></td>
</tr>
<tr>
<td>Bournemouth</td>
<td></td>
</tr>
</tbody>
</table>

In the next figure, we plot the estimated abilities at the end of the season (i.e., after week 38). This time, we sort the teams in the tables according to their final standings in the table (i.e., sorted according to total points) at the end of the season. We observe that the estimated abilities are fairly consistent with the actual rankings and the uncertainty intervals are narrower compared to the results after week 1.
We also examine the evolution of abilities for each team over the course of the season. In Fig. 2, we plot the estimated abilities for all the teams after each week (using the data from matches upto and including that week). The uncertainty intervals (+/- 1 s.e.) are shown with gray bands. The score differences are shown by red dots. We observe that ability movements are consistent with score differences. For instance, Leicester did not do well in the previous season; so, its initial ability is not high. Despite some good results at the beginning of the season, they got some bad results in weeks 4 to 9 (a drop in the estimated ability). Afterwards, they did very well until the end of the season which is shown by the positive trend in the estimated abilities.

Figure 1: Leicester performance and position in the table (source: Wikipedia)
Figure 2: For each of the 20 teams in the division, the black line shows estimated ability as the season goes on (at each point, posterior mean +/- 1 s.e. given only the first w weeks of the season, for each w); the red dots display the score difference for each game.
Parameter Estimates

The estimated model parameters after week 38 (all matches) are shown below. We observe that the home teams have an average of 0.28 goals per game advantage. Also, the small value for $\tau_a$ shows that the game-to-game variation is relatively similar for all teams. The large value of estimated $\nu$ indicates that we could replace the $t$-student distribution with the normal without much change.

<table>
<thead>
<tr>
<th></th>
<th>mean</th>
<th>se_mean</th>
<th>sd</th>
<th>2.5%</th>
<th>25%</th>
<th>50%</th>
<th>75%</th>
<th>97.5%</th>
<th>n_eff</th>
<th>Rhat</th>
</tr>
</thead>
<tbody>
<tr>
<td>b_home</td>
<td>0.28</td>
<td>0.002</td>
<td>0.08</td>
<td>1e-01</td>
<td>0.22</td>
<td>0.28</td>
<td>0.3</td>
<td>0.4</td>
<td>1500</td>
<td>1</td>
</tr>
<tr>
<td>b_prev</td>
<td>0.59</td>
<td>0.006</td>
<td>0.18</td>
<td>3e-01</td>
<td>0.47</td>
<td>0.59</td>
<td>0.7</td>
<td>0.9</td>
<td>963</td>
<td>1</td>
</tr>
<tr>
<td>sigma_a0</td>
<td>0.28</td>
<td>0.008</td>
<td>0.14</td>
<td>3e-02</td>
<td>0.19</td>
<td>0.29</td>
<td>0.4</td>
<td>0.6</td>
<td>285</td>
<td>1</td>
</tr>
<tr>
<td>tau_a</td>
<td>0.08</td>
<td>0.002</td>
<td>0.04</td>
<td>8e-03</td>
<td>0.05</td>
<td>0.08</td>
<td>0.1</td>
<td>0.2</td>
<td>372</td>
<td>1</td>
</tr>
<tr>
<td>nu</td>
<td>26.40</td>
<td>0.346</td>
<td>13.39</td>
<td>1e+01</td>
<td>16.34</td>
<td>23.62</td>
<td>33.0</td>
<td>59.7</td>
<td>1500</td>
<td>1</td>
</tr>
<tr>
<td>sigma_y</td>
<td>1.47</td>
<td>0.002</td>
<td>0.07</td>
<td>1e+00</td>
<td>1.43</td>
<td>1.48</td>
<td>1.5</td>
<td>1.6</td>
<td>1500</td>
<td>1</td>
</tr>
</tbody>
</table>

Model Checking

As part of the Stan model, we sample replicated data for `score_diff` in the `generated_quantities` block. We can then check whether the actual score differences are consistent with the distribution of replicated data. Here, we examine the replicated score differences achieved by fitting the model to all the data (week 1 to week 38). In the figure below, all 380 matches are sorted according to their score difference (shown in black dots). For each match, the median of the replicated score differences is shown in red, the 95% uncertainty interval is shown in light yellow, and the 50% uncertainty interval is shown in dark yellow. We observe that most of the actual score differences are in the uncertainty intervals; in fact, 96.3% of them are in the interval. We can plot the same figure for the 50% uncertainty interval. In this case, 55.5% of the actual score differences are in the interval.
Making Probabilistic Predictions with The Model

We can use our estimates in week $w$ to predict matches in week $w + 1$. A part of the code for this is shown below. We use the parameters from our 1500 draws to simulate score differences from the posterior predictive distribution.

```r
rt_ls <- function(n, df, mu, a) rt(n, df)*a + mu
for (i in 11:380) {
  w <- ceiling(i/10)
  for (j in 1:nsamples) {
    score_diff_pred[i,j] <-
      rt_ls(1, nu[w-1,j],
            a_sims[j,epl$home_week[i]-1, epl$home_team[i]] -
            a_sims[j,epl$away_week[i]-1, epl$away_team[i]] +
            b_home[w-1,j],
            sigma_y[w-1,j]);
  }
}
```

We can then compare the distribution of predicted score differences with the actual score differences. We do the comparison in the second half of the season, allowing the model to see enough data before making predictions. The results are shown in the figure below. In this figure, all 190 matches in the second half of the season are sorted according to their score difference (shown in black dots). For each match, the median of the predicted score differences is shown in red, the 95% uncertainty interval is shown in light yellow, and the 50% uncertainty interval is shown in dark yellow. We observe that most of the actual score differences are in the uncertainty intervals of predictions; in fact, 94.7% of them are in the interval. We can plot the same figure for the 50% uncertainty interval. In this case, 53.2% of the actual score differences are in the interval.

![Image](image_url)
As part of our data, we have the betting odds offered for the matches. We can use this information to assess the quality of our predictions. For each match, we have probabilistic predictions—a distribution of predicted score differences. We can translate this information into a decision in a number of ways. Here, we use the median of the score difference and round it to the closest integer; depending on whether this value is positive, negative, or zero, we bet win, lose, or draw and wager $1. We plot our cumulative winnings below. The total winnings after removing wagers is $15.4 in the end—a return of 8.08%! 

![cumulative winnings graph](chart.png)